Fourth International Conference on
Geometry, Integrability and Quantization
June 6–15, 2002, Varna, Bulgaria
Ivaïlo M. Mladenov and Gregory L. Naber, Editors
Coral Press, Sofia 2003, pp 206–234

UNDULOIDS AND THEIR CLOSED GEODESICS

IVAÏLO MLADENOV† and JOHN OPREA‡

† *Institute of Biophysics, Bulgarian Academy of Sciences
Acad. G. Bonchev Street, Block 21, 1113 Sofia, Bulgaria*

‡ *Department of Mathematics, Cleveland State University
Cleveland OH, 44115 USA*

Abstract. We construct explicit parametrizations in terms of elliptic functions for unduloids depending on a parameter. We then use these parametrizations to study geodesics on unduloids. In particular, we use *Maple* to find interesting closed geodesics on unduloids.

1. Introduction

Finding closed geodesics on manifolds is an important and quite difficult problem in geometry (e. g. see [10]). For instance, it was a major advance when, around 1930, Lusternik and Schnirelmann showed that S^2 with any Riemannian metric possesses at least 3 closed geodesics. Recent acclaimed work of Franks [4] shows that, in fact, an infinite number of distinct closed geodesics exist. Lusternik and Fet proved that every closed manifold possesses at least one closed geodesic and this type of result can be extended for, say, surfaces not homeomorphic to the plane or cylinder [16]. In general, the focus has been on proving existence theorems rather than finding explicit examples of closed geodesics. In this paper, we will consider a non-compact cylinder-like surface called the *unduloid* and study some aspects of its geodesics. In particular, we will see that *Maple* may be used to find (within some tolerance) closed geodesics of an interesting type. We will *not* simply draw pictures and make claims, however, but rather, prove that our computer methods determine closed geodesics.(1) (2) For this, we bring classical differential geometry and

(1) We wish to thank David Singer for several conversations which contributed mightily to the development of the method for finding closed geodesics presented here.

(2) A preliminary version of this paper has appeared in the *Proceedings of Maple Workshop* 2002.

the subject of elliptic functions to bear.

We will begin by reviewing the subject of elliptic functions and the methods of differential geometry which pertain to geodesic determination. For a brief history of the development of elliptic functions, see [15]. For a straightforward exposition of their properties and applications, see [7] (as well as [3] and [9]). Finally, for a recent approach in terms of dynamical systems, see [11].

The basics of differential geometry (especially in conjunction with computer algebra systems) may be found in [12, 14, 6].

2. Recollections of Elliptic Functions

The easiest way to understand the elliptic functions is to consider them as analogues of the ordinary trigonometric functions. From freshman calculus, we know that

$$\arcsin(x) = \int_0^x \frac{\mathrm{d}u}{\sqrt{1-u^2}} \,.$$

Of course, if $x = \sin(t)$ $(-\pi/2 \leq t \leq \pi/2)$, then we have

$$t = \arcsin(\sin(t)) = \int_0^{\sin(t)} \frac{\mathrm{d}u}{\sqrt{1-u^2}} \,.$$

In this way, we may view $\sin(t)$ as an inverse function for the integral. Now, fixing some k with $0 \leq k \leq 1$ (called the **modulus**), we make the

Definition 1. *The **Jacobi sine function*** $\operatorname{sn}(u,k)$ *as the inverse function of the following integral. Namely,*

$$u = \int_0^{\operatorname{sn}(u,k)} \frac{\mathrm{d}t}{\sqrt{1-t^2}\sqrt{1-k^2t^2}} \,. \tag{1}$$

More generally, we write

$$F(z,k) = \int_0^z \frac{\mathrm{d}t}{\sqrt{1-t^2}\sqrt{1-k^2t^2}} \tag{2}$$

and call $F(z,k)$ *an **elliptic integral of the first kind**. An **elliptic integral of the second kind** is defined by*

$$E(z,k) = \int_0^z \frac{\sqrt{1-k^2t^2}}{\sqrt{1-t^2}} \,\mathrm{d}t \,.$$

When $z = 1$ in $F(z,k)$ and $E(z,k)$, then these integrals are respectively denoted by $K(k)$ and $E(k)$ and called the ***complete*** *elliptic integrals of the first and second kind.*

The **Jacobi cosine function** $\mathrm{cn}(u,k)$ may be defined in terms of $\mathrm{sn}(u,k)$

$$\mathrm{sn}^2(u,k) + \mathrm{cn}^2(u,k) = 1\,.$$

A third **Jacobi elliptic function** $\mathrm{dn}(u,k)$ is defined by the equation

$$\mathrm{dn}^2(u,k) + k^2\,\mathrm{sn}^2(u,k) = 1\,.$$

The integral definition of $\mathrm{sn}(u,k)$ makes it clear that, $\mathrm{sn}(u,0) = \sin(u)$. Of course, $\mathrm{cn}(u,0) = \cos(u)$ as well. The derivatives of the elliptic functions can be found from the definitions (but also see [11] where the elliptic functions are essentially defined in terms of their derivatives!). For instance, let us compute the derivative of $\mathrm{sn}(u,k)$. Suppose in (2) that $z = z(u)$. Then

$$\frac{\mathrm{d}F}{\mathrm{d}u} = \frac{\mathrm{d}F}{\mathrm{d}z}\,\frac{\mathrm{d}z}{\mathrm{d}u} = \frac{1}{\sqrt{1-z^2}\sqrt{1-k^2z^2}}\,\frac{\mathrm{d}z}{\mathrm{d}u}\,.$$

But, from (1), we know that, when $z = \mathrm{sn}(u,k)$, we have $F(z,k) = u$. Hence, replacing z by $\mathrm{sn}(u,k)$ and using $\mathrm{d}u/\,\mathrm{d}u = 1$, we obtain

$$\begin{aligned}
1 &= \frac{1}{\sqrt{1-\mathrm{sn}(u,k)^2}\sqrt{1-k^2\,\mathrm{sn}(u,k)^2}}\,\frac{\mathrm{d}\,\mathrm{sn}(u,k)}{\mathrm{d}u} \\
\frac{\mathrm{d}\,\mathrm{sn}(u,k)}{\mathrm{d}u} &= \sqrt{1-\mathrm{sn}(u,k)^2}\sqrt{1-k^2\,\mathrm{sn}(u,k)^2} \\
\frac{\mathrm{d}\,\mathrm{sn}(u,k)}{\mathrm{d}u} &= \mathrm{cn}(u,k)\,\mathrm{dn}(u,k)\,.
\end{aligned}$$

We also have

$$\begin{aligned}
\frac{\mathrm{d}\,\mathrm{cn}(u,k)}{\mathrm{d}u} &= -\,\mathrm{sn}(u,k)\,\mathrm{dn}(u,k) \\
\frac{\mathrm{d}\,\mathrm{dn}(u,k)}{\mathrm{d}u} &= -k^2\,\mathrm{sn}(u,k)\,\mathrm{cn}(u,k)\,.
\end{aligned} \tag{3}$$

Maple has built-in elliptic functions, so we can easily plot them as follows. We take $k = 1/\sqrt{2}$ for concreteness.

```
> with(plots):with(linalg):
```

We will need to have a numerical value for $K(1/\sqrt{2})$. We can find this as follows.

```
> k1 := 1/sqrt(2);
```

$$k1 := \frac{1}{2}\sqrt{2}$$

```
> fsolve(JacobiSN(u, k1) = 1, u);
```

$$1.854074677$$

Here are the plots of the elliptic sine sn, cosine cn and dn.

```
> plot({JacobiSN(u, k1), JacobiCN(u, k1), JacobiDN(u, k1)},
u = -1.854074677..3*1.854074677);
```



Figure 1. Elliptic sine, cosine and dn

We see from Fig. 1 that $\mathrm{sn}(u,k)$ and $\mathrm{cn}(u,k)$ are periodic, but what is the period? We can determine the period by recalling (1)

$$u = \int_0^{\mathrm{sn}(u,k)} \frac{\mathrm{d}t}{\sqrt{1-t^2}\sqrt{1-k^2t^2}}$$

and

$$K(k) = \int_0^1 \frac{\mathrm{d}t}{\sqrt{1-t^2}\sqrt{1-k^2t^2}} .$$

We see that, from the first equality, $\mathrm{sn}(K(k),k) = 1$. Clearly, from the graph, we find that $K(k)$ is $1/4$ of the period of $\mathrm{sn}(u,k)$. Of course, this can be verified analytically (see, for instance, [18, p. 368]), but this argument suffices for our purposes. Also, the identity, $\mathrm{sn}^2(u,k) + \mathrm{cn}^2(u,k) = 1$ implies that $\mathrm{cn}(u,k)$ has the same period as $\mathrm{sn}(u,k)$ and that $\mathrm{cn}(K(k),k) = 0$. So, the value of the complete elliptic integral $K(k)$ can be found also by calculation where the elliptic cosine vanishes.

Finally, we will require the following result later.

Proposition 1. *The following identity holds* [1]

$$\int_0^u \operatorname{cn}^2(\tilde{u},k)\,\mathrm{d}\tilde{u} = \frac{E(\operatorname{sn}(u,k),k)}{k^2} - \frac{1-k^2}{k^2}F(\operatorname{sn}(u,k),k)\,. \tag{4}$$

Proof: We compute from the definitions

$$\begin{aligned}\frac{E(\operatorname{sn}(u,k),k)}{k^2} - \frac{1-k^2}{k^2}F(\operatorname{sn}(u,k),k) \\ &= \int_0^{\operatorname{sn}(u,k)} \left(\frac{1}{k^2}\frac{\sqrt{1-k^2t^2}}{1-t^2} - \frac{(1-k^2)}{k^2}\frac{1}{\sqrt{(1-t^2)(1-k^2t^2)}}\right)\mathrm{d}t \\ &= \int_0^{\operatorname{sn}(u,k)} \frac{1-k^2t^2-(1-k^2)}{k^2\sqrt{1-t^2}\sqrt{1-k^2t^2}}\,\mathrm{d}t = \int_0^{\operatorname{sn}(u,k)} \frac{\sqrt{1-t^2}}{\sqrt{1-k^2t^2}}\,\mathrm{d}t\,.\end{aligned}$$

Now make the substitution $t = \operatorname{sn}(\tilde{u},k)$ with $\mathrm{d}t = \operatorname{cn}(\tilde{u},k)\operatorname{dn}(\tilde{u},k)\,\mathrm{d}\tilde{u}$ and note that the limits of integration transform as follows. When $t = 0$, we have $\operatorname{sn}(\tilde{u},k) = 0$, so $\tilde{u} = 0$. When $t = \operatorname{sn}(u,k)$, then $\operatorname{sn}(\tilde{u},k) = \operatorname{sn}(u,k)$, so $\tilde{u} = u$. Hence, we obtain

$$\begin{aligned}\frac{E(\operatorname{sn}(u,k),k)}{k^2} - \frac{1-k^2}{k^2}F(\operatorname{sn}(u,k),k) &= \int_0^u \frac{\operatorname{cn}(\tilde{u},k)\operatorname{cn}(\tilde{u},k)\operatorname{dn}(\tilde{u},k)\,\mathrm{d}\tilde{u}}{\operatorname{dn}(\tilde{u},k)} \\ &= \int_0^u \operatorname{cn}^2(\tilde{u},k)\,\mathrm{d}\tilde{u}\,.\end{aligned}$$

□

3. The Calculus of Variations

Elliptic integrals often arise in the calculus of variations (as we shall see below). The calculus of variations deals with problems of the following sort: find a curve $y = y(x)$ which makes the integral

$$J = \int_{x_0}^{x_1} F(x, y(x), y'(x))\,\mathrm{d}x$$

[1] Although $F(\operatorname{sn}(u_0,k),k) = u_0$, we shall continue to write $F(\operatorname{sn}(u_0,k),k)$ because simplification in *Maple* appears to work better in that case.

a minimum (or, more generally, a critical value). A (famous) necessary condition for a minimizer $y(x)$ is the **Euler–Lagrange equation**

$$\frac{\mathrm{d}}{\mathrm{d}x}\left(\frac{\partial F}{\partial y'}\right) - \frac{\partial F}{\partial y} = 0\,. \tag{5}$$

A curve $y(x)$ which satisfies the Euler–Lagrange equation is said to *extremize* J. A more compact notation for this equation is $\dfrac{\mathrm{d}}{\mathrm{d}x}(F_{y'}) - F_y = 0$. Two variations of the Euler–Lagrange necessary condition arise when we have the following situations:

i) If $F(x, y, y')$ does not explicitly depend on x, then the Euler–Lagrange equation may be replaced by the *first integral*

$$F - y'\frac{\partial F}{\partial y'} = c \tag{6}$$

where c is a constant. The equivalence of the first integral with the Euler–Lagrange equation is seen by simply differentiating the first integral with respect to x.

ii) If we wish to minimize $J = \int_{x_0}^{x_1} F(x, y(x), y'(x))\,\mathrm{d}x$ subject to an extra constraint $I = \int_{x_0}^{x_1} G(x, y, y')\,\mathrm{d}x$, then we take either the Euler–Lagrange equation or the first integral associated to the integral

$$K = \int\limits_{x_0}^{x_1} (F(x, y, y') + \lambda G(x, y, y'))\,\mathrm{d}x\,.$$

The constant λ is called a *Lagrange multiplier*. See [12] for details.

4. Some Differential Geometry

We are interested in explicitly describing geometric objects by parametrization in terms of elliptic functions. Of course, one reason we want to do this is that we have an array of classical tools with which to study such a parametrized surface. (Modern expositions of the subject can be found in, for instance, [1, 6, 8, 12] and [17].) A parametrized surface $\mathcal{S}$ is determined by its **first** and **second fundamental forms**:

$$\begin{aligned} I &= E\,\mathrm{d}u^2 + 2F\,\mathrm{d}u\,\mathrm{d}v + G\,\mathrm{d}v^2 \\ II &= L\,\mathrm{d}u^2 + 2M\,\mathrm{d}u\,\mathrm{d}v + N\,\mathrm{d}v^2 \end{aligned} \tag{7}$$

where the coefficients are given by

$$\begin{aligned} E &= E[u,v] = \mathbf{x}_u \cdot \mathbf{x}_u\,, & F &= F[u,v] = \mathbf{x}_u \cdot \mathbf{x}_v\,, \\ G &= G[u,v] = \mathbf{x}_v \cdot \mathbf{x}_v\,, & L &= L[u,v] = \mathbf{x}_{uu} \cdot \mathbf{n}, \\ M &= M[u,v] = \mathbf{x}_{uv} \cdot \mathbf{n}, & N &= N[u,v] = \mathbf{x}_{vv} \cdot \mathbf{n}\,. \end{aligned} \tag{8}$$

Here $\mathbf{n}$ is the unit normal vector to $\mathcal{S}$

$$\mathbf{n} = \mathbf{n}[u,v] = \frac{\mathbf{x}_u \times \mathbf{x}_v}{\|\mathbf{x}_u \times \mathbf{x}_v\|}\,. \tag{9}$$

Intuitively, the **metric coefficients** E, F and G describe the stretching necessary to map a piece of the plane up to the surface under the parametrization. As can be seen from the definition, the coefficients L, M and N of II have more to do with acceleration and, hence, curvature. Indeed, there are classical formulas which describe two types of curvatures at every point of the surface. These are the **Gauss** and **mean** (meaning "average") curvatures, denoted by K and H respectively. The formulas are

$$K = \frac{LN - M^2}{EG - F^2} \quad \text{and} \quad H = \frac{EN + GL - 2FM}{2(EG - F^2)}\,.$$

We will deal only with surfaces of revolution which have parametrizations of the general form (up to permutation of coordinates)

$$\mathbf{x}(u,v) = (g(u), h(u)\cos(v), h(u)\sin(v))\,.$$

It is easy to compute that, for such surfaces, we always have $F = 0 = M$, so the formulas for Gauss and mean curvature reduce accordingly.

The *Maple 7* procedures which calculate these geometric quantities are as follows. First, we have procedures for the dot product, norm and cross product.

```
> dp := proc(X, Y)
X[1]*Y[1] + X[2]*Y[2] + X[3]*Y[3];
end:

> nrm := proc(X) local ans;
ans := sqrt(dp(X, X));
simplify(combine(ans), radical, symbolic, trig);
end:

> xp := proc(X, Y)
local a, b, c;
a := X[2]*Y[3] - X[3]*Y[2];
b := X[3]*Y[1] - X[1]*Y[3];
c := X[1]*Y[2] - X[2]*Y[1];
[a, b, c];
end:
```

The next procedures compute the metric E, F and G, the unit normal of a parametrized surface and the coefficients of II, L, M and N.

```
> EFG := proc(X)
local Xu, Xv, E, F, G;
Xu := [diff(X[1], u), diff(X[2], u), diff(X[3], u)];
Xv := [diff(X[1], v), diff(X[2], v), diff(X[3], v)];
E := dp(Xu, Xu); F := dp(Xu, Xv); G := dp(Xv, Xv);
simplify([E, F, G], symbolic);
end:
```

```
> UN := proc(X)
local Xu, Xv, Z, s;
Xu := [diff(X[1], u), diff(X[2], u), diff(X[3], u)];
Xv := [diff(X[1], v), diff(X[2], v), diff(X[3], v)];
Z := xp(Xu, Xv);
s := nrm(Z);
simplify([Z[1]/s, Z[2]/s, Z[3]/s], symbolic);
end:
```

```
> lmn := proc(X)
local Xu, Xv, Xuu, Xuv, Xvv, U, l, m, n;
Xu := [diff(X[1], u), diff(X[2], u), diff(X[3], u)];
Xv := [diff(X[1], v), diff(X[2], v), diff(X[3], v)];
Xuu := [diff(Xu[1], u), diff(Xu[2], u), diff(Xu[3], u)];
Xuv := [diff(Xu[1], v), diff(Xu[2], v), diff(Xu[3], v)];
Xvv := [diff(Xv[1], v), diff(Xv[2], v), diff(Xv[3], v)];
U := UN(X);
l := dp(U, Xuu); m := dp(U, Xuv); n := dp(U, Xvv);
simplify([l, m, n], symbolic);
end:
```

Of course, we immediately get procedures for calculating Gauss and mean curvature as well.

```
> GK := proc(X)
local E, F, G, l, m, n, S, T;
S := EFG(X);
T := lmn(X);
E := S[1]; F := S[2]; G := S[3];
l := T[1]; m := T[2]; n := T[3];
simplify((l*n - m^2)/(E*G - F^2), symbolic);
end:
```

```
> MK := proc(X)
local E, F, G, l, m, n, S, T;
S := EFG(X);
T := lmn(X);
E := S[1]; F := S[2]; G := S[3];
l := T[1]; m := T[2]; n := T[3];
simplify((E*n + G*l - 2*F*m)/(2*E*G - 2*F^2), symbolic);
end:
```

But we are interested in understanding finer details of a surface. Namely, we shall be interested in understanding something about the geodesics on a surface. Intuitively, **geodesics** are the straight lines of a surface; the shortest distances between points. Any curve $\boldsymbol{\alpha}(t)$ which lies on a surface S parametrized by $\mathbf{x}(u,v)$ may be written as $\boldsymbol{\alpha}(t) = \mathbf{x}(u(t), v(t))$, with $u(t)$ and $v(t)$ determining the curve. We therefore have the following result which serves for us as both a definition and proposition.

Proposition 2. *Geodesics are completely determined as solutions of a set of second order differential equations (once initial conditions are specified) called the* ***geodesic equations****:*

$$u'' + \frac{E_u}{2E}u'^2 + \frac{E_v}{E}u'v' - \frac{G_u}{2E}v'^2 = 0 \tag{10}$$

$$v'' - \frac{E_v}{2G}u'^2 + \frac{G_u}{G}u'v' + \frac{G_v}{2G}v'^2 = 0\,. \tag{11}$$

Here we have taken $F = 0$ since this will always be true for the surfaces we consider. We will use *Maple* to solve these equations numerically and plot geodesics on our special surfaces. There is also a special feature about geodesics on a surface of revolution which will allow us to predict geodesic behavior and then verify it pictorially. This feature is called the **Clairaut relation**.

Definition 2. *Suppose a parametrization $\mathbf{x}(u,v)$ has metric coefficients E and G which only depend on the parameter u and $F = 0$. Then $\mathbf{x}(u,v)$ is said to be u-**Clairaut**.*

When $\mathbf{x}(u,v)$ is u-Clairaut, the geodesic equations take the form

$$u'' + \frac{E_u}{2E}u'^2 - \frac{G_u}{2E}v'^2 = 0 \qquad v'' + \frac{G_u}{G}u'v' = 0\,.$$

In general, for a u-Clairaut parametrization $\mathbf{x}(u,v)$ and a unit speed geodesic $\boldsymbol{\alpha}$, we can reduce the second geodesic equation quite easily to first order. The equation $v'' + \frac{G_u}{G}u'v' = 0$ becomes

$$\begin{aligned}
\frac{v''}{v'} &= -\frac{G_u}{G}u' \\
\int \frac{v''}{v'}\mathrm{d}t &= -\int \frac{G_u}{G}u'\mathrm{d}t \\
\ln v' &= -\ln G + c \\
v' &= \frac{c}{G}\,.
\end{aligned}$$

Now, a geodesic $\boldsymbol{\alpha}$ may be assumed to be unit geodesic, so we have $\boldsymbol{\alpha}' = \mathbf{x}_u u' + \mathbf{x}_v v'$ with

$$|\boldsymbol{\alpha}'|^2 = E{u'}^2 + G{v'}^2 = 1\,.$$

This is called the *unit speed relation*. If the expression for v' is inserted into the unit speed relation, we obtain

$$\begin{aligned} 1 &= E{u'}^2 + G{v'}^2 \\ 1 &= E{u'}^2 + G\frac{c^2}{G^2} \\ 1 &= E{u'}^2 + \frac{c^2}{G} \\ {u'}^2 &= \frac{G - c^2}{EG} \\ u' &= \pm\sqrt{\frac{G - c^2}{EG}}\,. \end{aligned}$$

Now if we divide v' by u', we obtain an integral relating u and v.

Proposition 3. *For a u-Clairaut parametrization, geodesics are characterized by the integral relation*

$$v = \pm\int \frac{c\sqrt{E}}{\sqrt{G}\sqrt{G - c^2}}\,\mathrm{d}u\,.$$

Proof:

$$\begin{aligned} \frac{\mathrm{d}v}{\mathrm{d}u} &= \frac{v'}{u'} = \frac{\frac{c}{G}}{\pm\sqrt{\frac{G-c^2}{EG}}} = \frac{\pm c\sqrt{E}}{\sqrt{G}\sqrt{G - c^2}} \\ v &= \pm\int \frac{c\sqrt{E}}{\sqrt{G}\sqrt{G - c^2}}\mathrm{d}u\,. \end{aligned}$$

□

Let ϕ be the angle between the tangent vector of a unit speed geodesic $\boldsymbol{\alpha}$ and $\mathbf{x}_u$, the tangent vector of the u-parameter curve given by fixing a v-value in the parametrization $\mathbf{x}(u, v)$. Then the Clairaut relation holds.

Proposition 4. *For a u-Clairaut parametrization, the following relation holds:*

$$\sqrt{G}\sin\phi = c,\ \textit{where}\ \ c\ \textit{is the constant given above by}\ \ v' = \frac{c}{G}\,.$$

Proof:

$$\begin{aligned}\sin\phi &= \frac{\boldsymbol{\alpha}'\cdot\mathbf{x}_v}{|\boldsymbol{\alpha}'|\cdot|\mathbf{x}_v|} = \frac{(\mathbf{x}_u u' + \mathbf{x}_v v')\cdot\mathbf{x}_v}{1\cdot\sqrt{G}}\\ &= \frac{Gv'}{\sqrt{G}} = \sqrt{G}v' = \sqrt{G}\,\frac{c}{G}\\ \sqrt{G}\sin\phi &= c\,.\end{aligned}$$

□

We will see that the Clairaut relation restricts geodesics in fundamental ways. We can use *Maple 7* to calculate the geodesic equations, to solve them numerically and then to plot the solutions (i. e. geodesics) on a given parametrized surface.

```
> geoeq := proc(X)
local S, eq1, eq2;
S := EFG(X);
eq1 := diff(u(t),t$2) + subs({u = u(t), v = v(t)}, diff(S[1],u)/(2*S[1]))
*diff(u(t), t)^2 + subs({u = u(t), v = v(t)}, diff(S[1], v)/(S[1]))
*diff(u(t), t)*diff(v(t), t) - subs({u = u(t), v = v(t)},
diff(S[3], u)/(2*S[1]))*diff(v(t), t)^2 = 0;
eq2 := diff(v(t),t$2) - subs({u = u(t), v = v(t)}, diff(S[1],v)/(2*S[3]))
*diff(u(t), t)^2 + subs({u = u(t), v = v(t)}, diff(S[3], u)/(S[3]))
*diff(u(t), t)*diff(v(t), t) + subs({u = u(t), v = v(t)},
diff(S[3], v)/(2*S[3]))*diff(v(t), t)^2 = 0;
eq1, eq2;
end:
```

5. A Surface of Delaunay: The Unduloid

Suppose we have a surface of revolution which encloses a fixed volume $\mathcal{V} = \pi\int y(x)^2\,\mathrm{d}x$ such that surface area $A(\mathcal{S}) = 2\pi\int y(x)\sqrt{1+y'(x)^2}\,\mathrm{d}x$ is minimized. What are the resulting surfaces? Neglecting π in the formula, the set-up for this constrained problem takes the form

$$\text{extremize}\quad \int\left(2y(x)\sqrt{1+y'(x)^2} - \lambda y(x)^2\right)\mathrm{d}x\,.$$

Since the integrand does not depend on the independent variable x, we may use the first integral $f - y'(\partial f/\partial y') = \tilde{c}$ in place of the Euler–Lagrange equation to get

$$2y\sqrt{1+y'^2} - \lambda y^2 - y'\,\frac{2yy'}{\sqrt{1+y'^2}} = \tilde{c}\,.$$

We then obtain

$$y^2 + \frac{2ay}{\sqrt{1+y'^2}} = -c \tag{12}$$

where $a = -1/\lambda$ and $c = \tilde{c}/\lambda$. The surfaces of revolution determined by (12) are called **surfaces of Delaunay**. Surfaces of Delaunay were originally defined in [2] as surfaces of revolution of constant mean curvature. The variational viewpoint presented above is due to Sturm in an appendix to [2]. We can transform (12) into the following

$$\mathrm{d}x = -\frac{(y^2+c)\,\mathrm{d}y}{\sqrt{4a^2y^2-(y^2+c)^2}}\,. \tag{13}$$

Further on we will assume that the above constant c is strictly positive, that is $c = b^2$. The profile curves of surfaces of Delaunay may be characterized as those curves arising from rolling conics on a line. Such curves are called **roulettes** of conics. The case above with $c > 0$ corresponds to the roulette of an ellipse with axes a and b, $a > b$ (see [12]). These types of surfaces of Delaunay are called **unduloids**. The radicand in (13) is real just when y lies in the interval

$$\alpha = a - \sqrt{a^2-b^2}, \qquad \beta = a + \sqrt{a^2-b^2}\,. \tag{14}$$

Introducing the eccentricity ε of the ellipse,

$$\varepsilon = \sqrt{1-\frac{b^2}{a^2}} \tag{15}$$

the interval above can be rewritten in the form

$$\alpha = a(1-\varepsilon), \qquad \beta = a(1+\varepsilon)\,. \tag{16}$$

This leads us to change variables as follows:

$$y = a\sqrt{1+\varepsilon^2+2\varepsilon\sin\phi}, \qquad \phi \in [-\pi/2, \pi/2]\,. \tag{17}$$

Performing this change one gets

$$\mathrm{d}y \to \frac{a\varepsilon\cos\phi}{\sqrt{1+\varepsilon^2+2\varepsilon\sin\phi}}\,\mathrm{d}\phi, \qquad y^2+b^2 \to 2a^2(1+\varepsilon\sin\phi)$$

and

$$\sqrt{4a^2y^2-(y^2+b^2)^2} \to 2\varepsilon a^2\cos\phi\,.$$

As a result, the integral which we are interested in becomes

$$x(\phi) = -a \int_{-\pi/2}^{\phi} \frac{(1+\varepsilon \sin\tilde{\phi})}{\sqrt{1+\varepsilon^2+2\varepsilon\sin\tilde{\phi}}}\,\mathrm{d}\tilde{\phi} \tag{18}$$

and can be split further into two types of integrals which themselves can be transformed into elliptic integrals.

Lemma 1. *For* $k = \sqrt{\frac{2q}{p+q}}$, *let* $K = K(k)$ *and* $E(k)$ *denote the complete elliptic integrals of the first and second kind respectively, we have*

$$\int_{-\pi/2}^{\phi} \frac{\mathrm{d}\tilde{\phi}}{\sqrt{p+q\sin\tilde{\phi}}} = \frac{-2}{\sqrt{p+q}}(K + F(\operatorname{sn}(u,k),k))$$

and

$$\int_{-\pi/2}^{\phi} \frac{\sin\tilde{\phi}\,\mathrm{d}\tilde{\phi}}{\sqrt{p+q\sin\tilde{\phi}}} = -\frac{2\sqrt{p+q}}{q}(E(k)+E(\operatorname{sn}(u,k),k)) + \frac{2p}{q\sqrt{p+q}}(K+F(\operatorname{sn}(u,k),k))\,.$$

Here, u *corresponds to* ϕ *under the transformation* $\sin\tilde{\phi} = 1 - 2\operatorname{sn}^2(\tilde{u},k)$.

Proof: We make the substitution $\sin\tilde{\phi} = 1 - 2\operatorname{sn}^2(\tilde{u},k)$ with $\mathrm{d}\tilde{\phi} = -2\operatorname{dn}(\tilde{u})\,\mathrm{d}\tilde{u}$ (where, as usual, we shorten the notation for the elliptic functions for fixed k). Note that $\tilde{\phi} = -\pi/2$ corresponds to $-K$ under the substitution.

$$\begin{aligned}\int_{-\pi/2}^{\phi} \frac{\mathrm{d}\tilde{\phi}}{\sqrt{p+q\sin\tilde{\phi}}} &= \int_{-K}^{u} \frac{-2\operatorname{dn}(\tilde{u})}{\sqrt{p+q}\operatorname{dn}(\tilde{u})}\,\mathrm{d}\tilde{u} \\ &= \frac{-2}{\sqrt{p+q}}\int_{-K}^{u}\mathrm{d}\tilde{u} \\ &= \frac{-2}{\sqrt{p+q}}(K+F(\operatorname{sn}(u,k),k))\end{aligned}$$

where, again we use $F(\operatorname{sn}(u,k),k) = u$. For the second type of integral, we make the same substitution to obtain

$$
\begin{aligned}
&\int_{-\pi/2}^{\phi} \frac{\sin\tilde{\phi}}{\sqrt{p+q\sin\tilde{\phi}}}\,\mathrm{d}\tilde{\phi} \\
&\quad = \frac{-2}{\sqrt{p+q}} \int_{-K}^{u} \mathrm{d}\tilde{u} + \frac{4}{\sqrt{p+q}} \int_{-K}^{u} \mathrm{sn}^2(\tilde{u})\,\mathrm{d}\tilde{u} \\
&\quad = \frac{-2}{\sqrt{p+q}}(K+u) + \frac{4}{\sqrt{p+q}}(u+K) - \frac{4}{\sqrt{p+q}} \int_{-K}^{u} \mathrm{cn}^2(\tilde{u})\,\mathrm{d}\tilde{u} \\
&\quad = \frac{2}{\sqrt{p+q}}(K+u) - \frac{4}{\sqrt{p+q}} \left(\int_{-K}^{0} \mathrm{cn}^2(\tilde{u})\,\mathrm{d}\tilde{u} + \int_{0}^{u} \mathrm{cn}^2(\tilde{u})\,\mathrm{d}\tilde{u} \right) \\
&\quad = \frac{2}{\sqrt{p+q}}(K+u) - \frac{4}{\sqrt{p+q}} \left(\int_{0}^{K} \mathrm{cn}^2(\tilde{u})\,\mathrm{d}\tilde{u} + \int_{0}^{u} \mathrm{cn}^2(\tilde{u})\,\mathrm{d}\tilde{u} \right) \\
&\quad = \frac{2}{\sqrt{p+q}}(K+u) - \frac{4}{\sqrt{p+q}} \left(\frac{p+q}{2q} E(\mathrm{sn}(K)) - \frac{p-q}{2q} K \right. \\
&\qquad \left. + \frac{p+q}{2q} E(\mathrm{sn}(u)) - \frac{p-q}{2q} u \right) \\
&\quad = \frac{2}{\sqrt{p+q}}(K+u) - \frac{2}{q\sqrt{p+q}} ((p+q)E(k) - (p-q)K \\
&\qquad + (p+q)E(\mathrm{sn}(u)) - (p-q)u) \\
&\quad = \frac{2p}{q\sqrt{p+q}}(K+u) - \frac{2\sqrt{p+q}}{q} \left(E(k) + E(\mathrm{sn}(u),k) \right) .
\end{aligned}
$$

□

In our case, $p = 1 + \varepsilon^2$ and $q = 2\varepsilon$, so that

$$
k = \frac{2\sqrt{\varepsilon}}{1+\varepsilon} .
$$

Taken together, the above considerations give us

$$
x(u) = a(1-\varepsilon)[F(\mathrm{sn}(u,k),k) + K(k)] + a(1+\varepsilon)[E(\mathrm{sn}(u,k),k) + E(k)]
$$

where $K(k)$ and $E(k)$ are the complete elliptic integrals of the first and second kind, respectively. We must also convert $y(\phi)$ to $y(u)$ by the transformation

$\sin\phi = 1 - 2\,\mathrm{sn}^2(u)$. This gives

$$\begin{aligned} y(u) &= a\sqrt{(1+\varepsilon)^2 - 4\varepsilon\,\mathrm{sn}^2(u)} \\ &= a(1+\varepsilon)\sqrt{1 - \frac{4\varepsilon}{(1+\varepsilon)^2}\,\mathrm{sn}^2(u)} \\ &= a(1+\varepsilon)\sqrt{1 - k^2\,\mathrm{sn}^2(u)} = a(1+\varepsilon)\,\mathrm{dn}(u)\,. \end{aligned}$$

Revolving the curve $(x(u), y(u))$ around the x-axis leads to

Theorem 1. *A parametrization for the surface of Delaunay $\mathcal{S}$ is given by*

$$\mathbf{x}(u,v) = (x(u), y(u)\cos v, y(u)\sin v)$$

where

$$x(u) = a(1-\varepsilon)[F(\mathrm{sn}(u,k),k) + K(k)] + a(1+\varepsilon)[E(\mathrm{sn}(u,k),k) + E(k)]$$

and

$$y(u) = a(1+\varepsilon)\,\mathrm{dn}(u).$$

In *Maple*, we have

```
> delau := [a*(1 - epsilon)*(EllipticK(kk) + EllipticF(JacobiSN(u, kk), kk))
+ a*(1 + epsilon)*(EllipticE(kk) + EllipticE(JacobiSN(u, kk), kk)),
a*(1 + epsilon)*JacobiDN(u, kk)*cos(v),
a*(1 + epsilon)*JacobiDN(u, kk)*sin(v)];
```

The metric coefficients of the unduloid may be calculated by *Maple*.

```
> EFG(subs(kk = 2*sqrt(epsilon)/(1 + epsilon), delau));
```

$$-4\,\frac{\left[-\varepsilon^2 + 4\varepsilon\,\mathrm{JacobiSN}\left(u, 2\,\frac{\sqrt{\varepsilon}}{1+\varepsilon}\right)^2 - 2\varepsilon - 1\right] a^2}{1 + 2\varepsilon + \varepsilon^2},$$

$$0,$$

$$-\left[-\varepsilon^2 + 4\varepsilon\,\mathrm{JacobiSN}\left(u, 2\,\frac{\sqrt{\varepsilon}}{1+\varepsilon}\right)^2 - 2\varepsilon - 1\right] a^2$$

Using the relationships among the elliptic functions, we can simplify the results to

$$E = 4a^2\,\mathrm{dn}^2(u,k), \qquad F = 0, \qquad G = a^2(1+\varepsilon)^2\,\mathrm{dn}^2(u,k)\,.$$

Now let's use *Maple* to calculate the coefficients of the second fundamental form and the mean curvature. We write the results of the procedure "lmn" in standard LaTeX form because the Maple output is difficult to typeset.

```
> lmn(subs(kk = 2*sqrt(epsilon)/(1 + epsilon), delau));
```

$$\frac{4a\varepsilon\,\mathrm{dn}(u,2\frac{\sqrt{\varepsilon}}{1+\varepsilon})\,\mathrm{cn}(u,2\frac{\sqrt{\varepsilon}}{1+\varepsilon})\left(2\,\mathrm{sn}(u,2\frac{\sqrt{\varepsilon}}{1+\varepsilon})^2-1-\varepsilon\right)\sqrt{1+2\varepsilon+\varepsilon^2-4\varepsilon\,\mathrm{sn}(u,2\frac{\sqrt{\varepsilon}}{1+\varepsilon})^2}}{\left(-1-2\varepsilon-\varepsilon^2+4\varepsilon\,\mathrm{sn}(u,2\frac{\sqrt{\varepsilon}}{1+\varepsilon})^2\right)\sqrt{1-\mathrm{sn}(u,2\frac{\sqrt{\varepsilon}}{1+\varepsilon})^2}(1+\varepsilon)}$$

$$0$$

$$\frac{-\,\mathrm{cn}(u,2\frac{\sqrt{\varepsilon}}{1+\varepsilon})\,\mathrm{dn}(u,2\frac{\sqrt{\varepsilon}}{1+\varepsilon})\,a\left(2\varepsilon\,\mathrm{sn}(u,2\frac{\sqrt{\varepsilon}}{1+\varepsilon})^2+2\varepsilon^2\,\mathrm{sn}(u,2\frac{\sqrt{\varepsilon}}{1+\varepsilon})^2-2\varepsilon-\varepsilon^2-1\right)}{\sqrt{1+2\varepsilon+\varepsilon^2-4\varepsilon\,\mathrm{sn}(u,2\frac{\sqrt{\varepsilon}}{1+\varepsilon})^2}\sqrt{1-\mathrm{sn}(u,2\frac{\sqrt{\varepsilon}}{1+\varepsilon})^2}}$$

These formulas (with $k=2\sqrt{\varepsilon}/(1+\varepsilon)$ understood) reduce to

$$L=\frac{4a\varepsilon}{1+\varepsilon}\left(1-\frac{2}{1+\varepsilon}\,\mathrm{sn}(u)^2\right),\qquad M=0$$
$$N=a(1+\varepsilon)\left(1-\frac{2\varepsilon}{1+\varepsilon}\,\mathrm{sn}(u)^2\right).$$

From these and the metric coefficients, we can easily compute the mean curvature. *Maple* gives the following verification.

```
> MK(subs(kk = 2*sqrt(epsilon)/(1 + epsilon), delau));
```

$$\frac{1}{2a}\,\frac{(1+\varepsilon)\,\mathrm{JacobiCN}\left(u,2\frac{\sqrt{\varepsilon}}{1+\varepsilon}\right)\mathrm{JacobiDN}\left(u,2\frac{\sqrt{\varepsilon}}{1+\varepsilon}\right)}{\sqrt{1-\mathrm{JacobiSN}\left(u,2\frac{\sqrt{\varepsilon}}{1+\varepsilon}\right)^2}\sqrt{1+2\varepsilon+\varepsilon^2-4\varepsilon\,\mathrm{JacobiSN}\left(u,2\frac{\sqrt{\varepsilon}}{1+\varepsilon}\right)^2}}$$

This reduces to $H=1/(2a)$ and verifies that, in fact, unduloids (as all Delaunay surfaces) have constant mean curvature.

6. Geodesics on Unduloids

The metric coefficients of the parametrization of the unduloid were found to be

$$E=4a^2\,\mathrm{dn}^2(u),\quad F=0,\quad G=a^2(1+\varepsilon)^2\,\mathrm{dn}^2(u)\,.$$

These only depend on u (as they must since the unduloid is a surface of revolution), so the parametrization is u-Clairaut. Therefore, a Clairaut relation holds:

$$\sqrt{G}\sin(\theta)=a(1+\varepsilon)\,\mathrm{dn}(u)\sin(\theta)=c\,.$$

Geodesics on the unduloid obey the Clairaut relation. Thus, their behavior can be predicted. Moreover, Proposition 3 says that geodesics are characterized by the integral

$$
\begin{aligned}
v &= \pm \int \frac{c\sqrt{E}}{\sqrt{G}\sqrt{G-c^2}}\,\mathrm{d}u \\
&= \pm \frac{2\,\mathrm{dn}(u_0)}{(1+\varepsilon)} \int\limits_{u_0}^{u} \frac{\mathrm{d}u}{\sqrt{\mathrm{dn}^2(u)-\mathrm{dn}^2(u_0)}} \\
&= \pm \frac{\mathrm{dn}(u_0)}{\sqrt{\varepsilon}} \int\limits_{u_0}^{u} \frac{\mathrm{d}u}{\sqrt{\mathrm{sn}^2(u_0)-\mathrm{sn}^2(u)}}\,.
\end{aligned}
$$

Here, u_0 is the "initial" point of the geodesic and we have chosen $\phi = \pi/2$ in the Clairaut relation to get $c = a(1+\varepsilon)\,\mathrm{dn}(u_0)$ which has been substituted in the general integral formula. *Note* that v is continuous and monotonically increasing in u. This explains why our geodesics always proceed in a "forward" direction.

Approach 1. *The following steps determine closed geodesics on unduloids.*

- *Write a Maple procedure with inputs ε (to specify an unduloid) and u_0 (to specify a starting point for a geodesic which starts parallel to a parallel circle of the unduloid) and output the u-value corresponding to when the geodesic achieves revolution variable v equal to π.*
- We look for those outputs u which are zero. *Because we use numerical solutions of the geodesic equations to compute the endpoints of geodesics, we cannot hope to obtain $u = 0$. Rather,* because v is continuous and monotonically increasing, we look for positive and negative outputs *in order to guarantee a starting u_0 with final value $u = 0$ at $v = \pi$.*
- *The process above determines a closed geodesic on the unduloid for the following reasons. First, as we will see, in the parametrization of the unduloid, $u = 0$ corresponds to the "equatorial" geodesic depicted in Fig. 5. The unduloid is symmetric with respect to this equator, so, if we know that the u_0 geodesic travels π around the unduloid at $u = 0$, then symmetry says that it will travel another π until it reaches the point on the unduloid corresponding to $-u_0$. Then it will bounce off the parallel circle at $-u_0$ and travel around by a total of 2π back to the original parallel circle. Again, symmetry says that the geodesic must close up.*[(1)]

Remark 1. *We note here that it is not true in general that the endpoints of geodesics depend continuously on initial conditions. The fact that it is true here is due to the fact that the parametrization is u-Clairaut.*

[(1)] *Note that the geodesic obeys the Clairaut relation and it is known that a geodesic can never be asymptotic to a non-geodesic parallel circle. Thus, the geodesic must bounce back.*



Figure 2. An unduloid with $\varepsilon = 0.7$



Figure 3. An unduloid with $\varepsilon = 0.3$

Now let's use *Maple 7* to find geodesics according to the plan just enunciated.[(1)] First, to get a picture of various unduloids, we create a procedure to draw unduloids given inputs a and ε in the parametrization for unduloids above. See Fig. 2 and Fig. 3.

```
> with(plots):
```

[(1)] At this time, *Maple 8* is just becoming available. If necessary, updates to the procedures given below will appear on [13].

```
> plotund := proc(a, eps)
local kk, uu;
kk := 2*sqrt(eps)/(1 + eps);
uu := fsolve(JacobiSN(u, kk) = 1, u);
plot3d(subs({u = u(t), v = v(t)}, [a*(1 - eps)*(EllipticK(kk) +
EllipticF(JacobiSN(u, kk), kk)) + a*(1 + eps)*(EllipticE(kk) +
EllipticE(JacobiSN(u, kk), kk)), a*(1 + eps)*JacobiDN(u, kk)*cos(v),
a*(1 + eps)*JacobiDN(u, kk)*sin(v)]), u = -uu..uu, v = 0..2*Pi,
scaling = constrained, axes = boxed);
end:
```

```
> plotund(1, 0.7);
```

```
> plotund(1, 0.3);
```

Geodesics on the unduloid may be found by solving the geodesic equations numerically using "dsolve" and then putting the numerical solution values into the parametrization for the unduloid. This is the content of the following procedure. The inputs include the ε of the parametrization, initial points and derivatives, a running time "T", the number of points used to plot each geodesic "N", a grid of the form $[a, b]$, the orientation of the plot given by "theta" and "phi" and the color of the geodesic.

```
> undugeo := proc(eps, u0, v0, Du0, Dv0, T, N, gr, theta, phi, col)
local kk2, del, del0, ulim, desys, u1, v1, geo, plotX, equator;
kk2 := evalf(2*sqrt(eps)/(1 + eps));
ulim := fsolve(JacobiSN(u, kk2) = 1, u);
del := subs(a = 1, [a*(1 - eps)*(EllipticK(kk2)
+ EllipticF(JacobiSN(u, kk2), kk2))
+ a*(1 + eps)*(EllipticE(kk2) + EllipticE(JacobiSN(u, kk2), kk2)),
a*(1 + eps)*JacobiDN(u, kk2)*cos(v),
a*(1 + eps)*JacobiDN(u, kk2)*sin(v)]);
del0 := subs({a = 1, u = 0},
[a*(1 - eps)*(EllipticK(kk2) + EllipticF(JacobiSN(u, kk2), kk2))
+ a*(1 + eps)*(EllipticE(kk2) +
EllipticE(JacobiSN(u, kk2), kk2)), a*(1 + eps)*JacobiDN(u, kk2)*cos(v),
a*(1 + eps)*JacobiDN(u, kk2)*sin(v)]);
desys := dsolve({geoeq(del), u(0) = u0, v(0) = v0, D(u)(0) = Du0,
D(v)(0) = Dv0}, {u(t), v(t)}, type = numeric, output = listprocedure);
u1 := subs(desys, u(t)); v1 := subs(desys, v(t));
geo := tubeplot(subs(u = 'u1'(t), v = 'v1'(t), del), t = 0..T,
radius = 0.02, color = col, numpoints = N):
equator := tubeplot(del0, v = 0..2*Pi, radius = 0.015, color = black);
plotX := plot3d(subs({u = u(t), v = v(t)}, del), u = -ulim..ulim,
v = 0..2*Pi, grid = [gr[1], gr[2]], shading = ZGRAYSCALE):
display({geo, plotX, equator}, style = wireframe, scaling = constrained,
orientation = [theta, phi]);
end:
```

The following example plots a geodesic on a cylinder (i. e. the case $\varepsilon = 0$) which starts parallel to a parallel circle (see Fig. 4). In general, throughout this paper, we will start our geodesics parallel to a parallel circle. This is described by the initial conditions $Du0 = 0$, $Dv0 = 1$.

```
> undugeo(0.0, 0.4, 0, 0, 1, 20, 40, [20, 20], -109, 66, blue);
```



Figure 4. A geodesic on a cylinder

Of course, we shall mostly be concerned with $\varepsilon \neq 0$. In this case, a geodesic (which we hope will be closed) must bounce back and forth between parallel circles. The following procedure determines the u-value when the geodesic has gone around the unduloid by π. If $u = 0$, then symmetry considerations provide a closed geodesic. Because a numerical solution cannot be trusted to be $100\,\%$ accurate, we need to use something like monotonicity and continuity to guarantee a closed geodesic.

```
> halfbouncepoint := proc(u0, eps)
local kk2, del, geosystem, uuu, vvv, ttt, uuu0;
kk2 := 2*sqrt(eps)/(1 + eps);
del := subs(a = 1, [a*(1 - eps)*(EllipticK(kk2)
+ EllipticF(JacobiSN(u, kk2), kk2))
+ a*(1 + eps)*(EllipticE(kk2) + EllipticE(JacobiSN(u, kk2), kk2)),
a*(1 + eps)*JacobiDN(u, kk2)*cos(v), a*(1 + eps)*JacobiDN(u, kk2)*sin(v)]);
geosystem := dsolve({geoeq(del), u(0) = u0, v(0) = 0, D(u)(0) = 0,
D(v)(0) = 1},
{u(t), v(t)}, type = numeric, output = listprocedure, range = 0..25);
uuu := subs(geosystem, u(t)); vvv := subs(geosystem, v(t));
ttt := timelimit(30, fsolve(vvv(t) = evalf(Pi), t));
if type(ttt, float) then
uuu0 := eval(uuu(t), t = ttt);
uuu0;
else print('time error');
end if;
end:
```

The following example with $\varepsilon = 0.3$ shows our method. The first application of "halfbouncepoint" shows the existence of a closed geodesic. Further applications zoom in on the correct value for $u0$ and the closed geodesic is then plotted (see Fig. 6). Note that it is a general fact that a parallel circle is a geodesic where the profile curve has a critical point, so the equator (i. e. $u0 = 0$) on an unduloid is always a geodesic (see Fig. 5).



Figure 5. An equatorial geodesic



Figure 6. A closed geodesic for $\varepsilon = 0.3$

```
> for jj from -2 to 2 do
print(0.99 + jj*0.005, halfbouncepoint(0.99 + jj*0.005, 0.3)); od;
```

$$
\begin{array}{ll}
0.980, & -0.003546013760909292 38 \\
0.985, & -0.0015553716287617 8446 \\
0.99, & 0.00047813034775106 4930 \\
0.995, & 0.0025550465729754 8006 \\
1.000, & 0.0046759353601700 2660
\end{array}
$$

```
> for jj from -2 to 2 do
print(0.9888 + jj*0.0001, halfbouncepoint(0.9888 + jj*0.0001, 0.3)); od;
```

$$
\begin{array}{ll}
0.9886, & -0.0000956024228562558400 \\
0.9887, & -0.0000547339092084751702 \\
0.9888, & -0.0000138482941866551866 \\
0.9889, & 0.0000270544615188748633 \\
0.9890, & 0.0000679746262203292234
\end{array}
$$

```
> for jj from -2 to 2 do
print(0.9888338613 + jj*0.0000000001, halfbouncepoint(0.9888338613 +
jj*0.0000000001, 0.3)); od;
```

$$0.9888338611, \quad -0.215471530591940242 \times 10^{-9}$$
$$0.9888338612, \quad -0.131270777668085481 \times 10^{-9}$$
$$0.9888338613, \quad -0.470980860648589905 \times 10^{-10}$$
$$0.9888338614, \quad 0.371504283499385357 \times 10^{-10}$$
$$0.9888338615, \quad 0.121351875380024121 \times 10^{-9}$$

```
> undugeo(0.3, 0, 0, 0, 1, 25, 40, [20, 20], -119, 68, blue);
```

```
> undugeo(0.3, 0.9888338614, 0, 0, 1, 100, 200, [20, 20], -90, 68, blue);
```



Figure 7. A closed geodesic for $\varepsilon = 0.4$

Here is the case $\varepsilon = 0.4$. We show only the last step of the process of zooming in on $u0$ and the closed geodesic (or rather, the close approximation to the true one) in Fig. 7.

```
> for jj from -2 to 2 do
print(1.604921062 + jj*0.000000002, halfbouncepoint(1.604921062 +
jj*0.000000002, 0.4)); od;
```

$$1.604921058, \quad -0.975764764430883670 \times 10^{-8}$$
$$1.604921060, \quad -0.625177360401296678 \times 10^{-8}$$
$$1.604921062, \quad -0.276254281559792926 \times 10^{-8}$$
$$1.604921064, \quad 0.104294152827866249 \times 10^{-8}$$
$$1.604921066, \quad 0.454881944134469252 \times 10^{-8}$$

```
> undugeo(0.4, 1.604921062, 0, 0, 1, 30, 60, [20, 20], -109, 66, blue);
```

The next case shows that, as we get closer to a cylinder (i. e. as $\varepsilon \to 0$), closed geodesics of the kind we have depicted disappear.

```
> undugeo(0.25, 0.2, 0, 0, 1, 20, 40, [20, 20], -90, 66, blue);
```

```
> for jj from -2 to 2 do
print(0.25 + jj*0.03, halfbouncepoint(0.25 + jj*0.03, 0.25)); od;
```

$$0.19, \quad 0.000981709956794714302$$
$$0.22, \quad 0.00153040225539034531$$

$$0.25, \quad 0.002256501620956843844$$
$$0.28, \quad 0.003187349221511340444$$
$$0.31, \quad 0.004351508626454898844$$



Figure 8. A geodesic for $\varepsilon = 0.25$

We see that, as $u0$ increases, so does the u-value where $v = \pi$. We try to go lower to find the closed geodesic, but we never seem to get the negatives and positives which ensure the existence of a closed geodesic.

```
> for jj from -2 to 2 do
print(0.15 + jj*0.02, halfbouncepoint(0.15 + jj*0.02, 0.25)); od;
```

$$0.11, \quad 0.0001889461753438706144$$
$$0.13, \quad 0.0003123948928720046494$$
$$0.15, \quad 0.0004808193334321932664$$
$$0.17, \quad 0.0007014620705093029594$$
$$0.19, \quad 0.0009817099567947143024$$

```
> for jj from -2 to 2 do
print(0.06 + jj*0.03, halfbouncepoint(0.06 + jj*0.03, 0.25)); od;
```

$$0.00, \quad 0.0$$
$$0.03, \quad 0.383093716991672563 \times 10^{-5}$$
$$0.06, \quad 0.00003058323944613617904$$
$$0.09, \quad 0.0001033532751019917364$$
$$0.12, \quad 0.0002454961535499675504$$

Although we cannot seem to find a closed geodesic for $\varepsilon = 0.25$, we *can* find one for $\varepsilon = 0.27$.

```
> for jj from -2 to 2 do
print(0.647237683 + jj*0.000000002, halfbouncepoint(0.647237683 +
jj*0.000000002, 0.27)); od;
```

$$0.647237679, \quad -0.605163563574886988 \times 10^{-9}$$
$$0.647237681, \quad -0.236546335904280758 \times 10^{-9}$$
$$0.647237683, \quad 0.132067925606021986 \times 10^{-9}$$

$$0.647237685, \quad 0.235248143664809618 \times 10^{-9}$$
$$0.647237687, \quad 0.603848639060128312 \times 10^{-9}$$

```
> undugeo(0.27, 0.647237683, 0, 0, 1, 30, 60, [20, 20], -90, 68, red);
```



Figure 9. A closed geodesic for $\varepsilon = 0.27$

The procedure "halfbouncepoint" can produce unexpected results. For $\varepsilon = 0.45$, we actually find *two* closed geodesics by observing where changes of signs occur in the output of "halfbouncepoint".

```
> for jj from -2 to 2 do
print(1.82 + jj*0.02, halfbouncepoint(1.82 + jj*0.02, 0.45)); od;
```

$$1.78, \quad -0.116912442366751828$$
$$1.80, \quad -0.0682354690554627247$$
$$1.82, \quad -0.0155980583680302554$$
$$1.84, \quad 0.0411104550678680000$$
$$1.86, \quad 0.101949580950035340$$

```
> for jj from -2 to 2 do
print(1.8256479 + jj*0.000000004, halfbouncepoint(1.8256479 +
jj*0.000000004, 0.45)); od;
```

$$1.825647892, \quad 0.253827999504435686 \times 10^{-7}$$
$$1.825647896, \quad -0.110532436264562661 \times 10^{-6}$$
$$1.8256479, \quad -0.161765549514664153 \times 10^{-7}$$
$$1.825647904, \quad -0.254823376785034972 \times 10^{-6}$$
$$1.825647908, \quad 0.319462907100165338 \times 10^{-7}$$

Notice that we have two changes of sign here. In the following, we focus on both and plot the corresponding geodesics (see Fig. 10 and Fig. 11).

```
> for jj from -2 to 2 do
print(1.825647894 + jj*0.000000001, halfbouncepoint(1.825647894 +
jj*0.000000001, 0.45)); od;
```

$$1.825647892, \quad 0.253827999504435686 \times 10^{-7}$$
$$1.825647893, \quad 0.151038128462510635 \times 10^{-7}$$
$$1.825647894, \quad -0.579997126212011826 \times 10^{-7}$$
$$1.825647895, \quad -0.251446704209153452 \times 10^{-6}$$
$$1.825647896, \quad -0.110532436264562661 \times 10^{-6}$$

```
> undugeo(0.45, 1.8256478935, 0, 0, 1, 30, 60, [22, 20], -134, 79, red);
```



Figure 10. A closed geodesic for $\varepsilon = 0.45$



Figure 11. Another closed geodesic for $\varepsilon = 0.45$

```
> for jj from -2 to 2 do
print(1.825647906 + jj*0.000000001, halfbouncepoint(1.825647906 +
jj*0.000000001, 0.45)); od;
```

$$1.825647904, \quad -0.254823376785034972 \times 10^{-6}$$
$$1.825647905, \quad -0.227405424569997184 \times 10^{-6}$$
$$1.825647906, \quad 0.763889081985096030 \times 10^{-7}$$
$$1.825647907, \quad 0.377853410721620820 \times 10^{-7}$$
$$1.825647908, \quad 0.319462907100165338 \times 10^{-7}$$

```
> undugeo(0.45, 1.8256479055, 0, 0, 1, 30, 60, [22, 20], -112, 73, blue);
```

We can find closed geodesics up to a certain point. For example, for large ε's, we have something like the following.

```
> for jj from -2 to 2 do
print(3.065 + jj*0.001, halfbouncepoint(3.065 + jj*0.001, 0.75)); od;
```

$$3.063, \quad -0.05302973854710273\,11$$
$$3.064, \quad -0.03142151568199026\,00$$
$$3.065, \quad -0.009694233142784631\,36$$
$$3.066, \quad 0.01213910268453347\,22$$
$$3.067, \quad 0.03406851350562653\,54$$

```
> undugeo(0.75, 3.065, 0, 0, 1, 56, 70, [20, 20], -90, 68, red);
```



Figure 12. A closed geodesic for $\varepsilon = 0.75$



Figure 13. A wild geodesic for $\varepsilon = 0.85$

But, again, unusual behavior is observed after some point for larger ε's. Let's look at $\varepsilon = 0.85$. We have observed that the $u0$'s which give closed geodesics increase as ε increases. If we search for $u0$'s that work, we find the behavior below typical. Although it appears that Fig. 13 may be closed, further calculations show that this is not the case (as "halfbouncepoint" indicates).

```
> for jj from -2 to 2 do
print(3.85 + jj*0.02, halfbouncepoint(3.85 + jj*0.02, 0.85)); od;
```

$$3.81,\quad 3.09367094439529476$$
$$3.83,\quad 3.31473932779440883$$
$$3.85,\quad 3.53072473699172917$$
$$3.87,\quad 3.74103649971058516$$
$$3.89,\quad 3.88261887508628200$$

```
> undugeo(0.85, 3.85, 0, 0, 1, 100, 150, [20, 20], -137, 79, red);
> undugeo(0.85, 3.8, 0, 0, 1, 110, 150, [20, 20], -26, -108, red);
> undugeo(0.95, 3.85, 0, 0, 1, 160, 220, [20, 20], -137, 79, red);
```



Figure 14. Another wild geodesic for $\varepsilon = 0.85$



Figure 15. A wild geodesic for $\varepsilon = 0.95$

7. Open Problems

The central question which must be addressed in the future is precisely how the existence of closed geodesics (of the type we are interested in) depends on ε. In particular, the following problems should be considered:

- Find the exact $\varepsilon < 0.27$ where closed geodesics (of the type above) don't exist.
- Use the integral for v to analyze the cutoff $\varepsilon < 0.27$ where closed geodesics (of the type above) don't exist.
- Modify the procedure "halfbouncepoint" to handle the cases $\varepsilon > 0.75$ where geodesics become very complicated. Do closed geodesics exist?
- Of course, we have concentrated on closed geodesics of a very particular type. The same arguments as given above would guarantee a *closed* geodesic (with many self-intersections) if one of our "parallel" starting geodesics hits the equator at any rational multiple of π. It is very easy to modify "halfbouncepoint" to handle this case, but it is unclear what the correct guesses should be to even begin a systematic search. Nevertheless, for the ε where we have not found closed geodesics of simple type, the rational multiple idea may yet yield more complicated closed geodesics.

References

[1] Berger M. and Gostiaux B., *Differential Geometry: Manifolds, Curves and Surfaces*, Springer, New York, 1988.

[2] Delaunay C., *Sur la surface de revolution dont la courbure moyenne est constante*, J. Math. Pures et Appliquées **6** (1841) 309–320.

[3] Dwight H., *Tables of Integrals and Other Mathematical Data*, Fourth Edition, Macmillan, New York 1961.

[4] Franks J., *Geodesics on S^2 and Periodic Points of Annulus Homeomorphisms*, Invent. Math. **108** (1992) 403–418.

[5] Gradshteyn I. and Ryzhik I., *Tables of Integrals, Series and Products*, Academic Press, 1980.

[6] Gray A., *Modern Differential Geometry of Curves and Surfaces with Mathematica®*. Second Edition, CRC Press, Boca Raton 1998.

[7] Greenhill A., *The Applications of Elliptic Functions*, Macmillan, London 1892.

[8] Henderson D., *Differential Geometry: A Geometric Introduction*, Prentice Hall, New Jersey 1998.

[9] Janhke E., Emde F. and Lösch F., *Tafeln Höherer Funktionen*, Teubner Verlag, Stuttgart 1960.

[10] Klingenberg W., *Lectures on Closed Geodesics*, Springer 1978 p. 230.

[11] Meyer K., *Jacobi Elliptic Functions from a Dynamical Systems Point of View*, Amer. Math. Monthly **108** (2001) 729–737.

[12] Oprea J., *Differential Geometry and Its Applications*, Second Edition, Prentice Hall, New Jersey 2002.

[13] Oprea J., www.csuohio.edu/math/oprea.

[14] Oprea J., *The Mathematics of Soap Films: Explorations with Maple®*, AMS, Providence, Rhode Island 2000.

[15] Stillwell J., *Mathematics and its History*, Springer, New York 1989.

[16] Thorbergsson G., *Closed Geodesics on Non-compact Riemannian Manifolds*, Math. Z. **159** (1978) 249–258.

[17] Willmore T., *An Introduction to Differential Geometry*, Second Edition, Oxford Univ. Press, Oxford 1982.

[18] Woods F., *Advanced Calculus*, Ginn and Co. 1934.